
TITLE: 
Elfi00/Vfnca1/Vfcc00: explosion in a 2D deformable container with an internal 
diaphragm. 
 
PROBLEM: 
The purpose is to compare Finite Element, Finite Volume (Node-Centred) and Finite 
Volume (Cell-Centred) solutions of this problem and to highlight the particularities of 
each model. 
  
MESH: 
The model is 2D plane strain and contains 20 x 10 finite elements (or finite volumes) 
in the fluid. The structural mesh is conforming to the fluid mesh. 
 
MATERIALS: 
The structure is an elasto-plastic VM23 material. The explosive and the gas (air) are 
rep;resented by perfect gases at high and low pressures, respectively. The two gases 
may mix up. 
 
BOUNDARY CONDITIONS: 
The fluid-structure interaction along the deformable container walls is automatically 
modelled by the FSA directive in the FE and N-C FV solutions. The C-C FV solution 
requires no specific directive and the fluid and structure nodes are merged together. 
 
LOADING: 
The system is initially at rest, but not in equilibrium. 

CALCULATION: 
The calculation is performed up to 5 ms. This time is sufficient to see propagation of 
the pressure waves to the top of the box and several reflections. 
 
RESULTS: 
Although no comparison with experiments can be done on this problem, the results 
appear physically correct. 
 
POST-TREATMENT 
Several animations of the computed results are presented. Time curves of the fluid 
pressures and structure displacements are compared. 
 
Numerical Solutions (Finite Elements) 
 
ELFI00 
 
The mesh generation file (K2000) is: 
 
opti echo 1; 
opti titr 'ELFI00'; 
opti sauv form 'elfi00.msh'; 
opti trac psc ftra 'elfi00_mesh.ps'; 
* 
opti dime 2 elem qua4; 
* 
p1 = 0 0; 
p2 = 2 0; 
p3 = 2 1; 
p4 = 0 1; 
p5 = 0.4 0; 
p6 = 0.4 0.4; 
p7 = 0 0.4; 
p8 = 0.4 1; 
* 
tol = 0.001; 
* 
c1 = p1 d 4 p5; 
c2 = p5 d 4 p6; 
c3 = p6 d 4 p7; 
c4 = p7 d 4 p1; 
expl = dall c1 c2 c3 c4 plan; 
* 
p9l = 1 0; 
p9r = 1 0; 
p10 = 1 0.4; 
p11 = 2 0.4; 
* 
c1 = p5 d 6 p9l; 
c2 = p9l d 4 p10; 
c3 = p10 d 6 p6; 
c4 = p6 d 4 p5; 
air1a = dall c1 c2 c3 c4 plan; 
* 
c1 = p9r d 10 p2; 
c2 = p2 d 4 p11; 
c3 = p11 d 10 p10; 
c4 = p10 d 4 p9r; 
air1b = dall c1 c2 c3 c4 plan; 
* 
c1 = p6 d 6 p10 d 10 p11; 
c2 = p11 d 6 p3; 
c3 = p3 d 16 p8; 
c4 = p8 d 6 p6; 
air1c = dall c1 c2 c3 c4 plan; 

* 
elim tol (air1a et air1c); 
elim tol (air1b et air1c); 
air1 = air1a et air1b et air1c; 
* 
c1 = p7 d 4 p6; 
c2 = p6 d 6 p8; 
c3 = p8 d 4 p4; 
c4 = p4 d 6 p7; 
air2 = dall c1 c2 c3 c4 plan; 
elim tol (air1c et air2); 
elim tol (expl et air2); 
elim tol (expl et air1a); 
* 
air = air1 et air2; 
flui = expl et air; 
* 
e1 = expl elem cont p1; 
e3 = air1 elem cont p3; 
* 
p1s = p1 plus p1; 
p2s = p2 plus p1; 
p3s = p3 plus p1; 
p4s = p4 plus p1; 
p9s = p9l plus p1; 
p10s = p10 plus p1; 
p12s = 1 1; 
c1s = p1s d 10 p9s d 10 p2s; 
c2s = p2s d 10 p3s; 
c3s = p3s d 10 p12s d 10 p4s; 
c4s = p4s d 10 p1s; 
c5s = p9s d 4 p10s; 
stru = c1s et c2s et c3s et c4s et c5s; 
* 
nfsa = cont flui; 
* 
mesh = flui et stru et e1 et e3 et nfsa; 
* 
sauv form mesh; 
trac qual mesh; 
* 
list (nbel mesh); 
list (nbno mesh); 
* 
fin; 

 
The EUROPLEXUS input file reads: 
 
ELFI00 
ECHO 
!conv win 
CAST mesh 
DPLA ALE 
DIME 
 PT2L 235 PT3L 64 FL24 200 ED01 64 ZONE 2 
 NALE 26 NBLE 167 
TERM 
GEOM FL24 flui ED01 stru TERM 
COMP EPAI 0.01 LECT stru TERM 
     COUL turq LECT air  TERM 
          roug LECT expl TERM 
          bleu LECT stru TERM 
GRIL LAGR LECT stru TERM 
     EULE LECT nfsa TERM 
     AUTO AUTR 
MATE FLUT RO 5.9485 EINT 4.20274E5 GAMM 1.4 PB 0 PREF 1.E5 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 
          LECT expl TERM 
     FLUT RO 1.1897 EINT 2.10137E5 GAMM 1.4 PB 0 PREF 1.E5 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 
          LECT air TERM 
     VM23 RO 7800. YOUNG 1.6E11 NU 0.333 ELAS 1.05E8 
          TRAC 2 1.05E8 .656256E-3 1.6105E10 1.00066 
          LECT stru TERM 
LINK COUP FSA LECT nfsa TERM 
ECRI DEPL VITE ACCE FINT FEXT CONT ECRO TFRE 1.E-3 
          POIN LECT p1 p5 p6 p7 p3 TERM 
          ELEM LECT e1 e3 TERM 
     FICH ALIC TFRE 5.E-5 
     FICH ALIC TEMP FREQ 1 
          POIN LECT p1 p5 p6 p7 p3 p1s p2s p3s p4s p9s p10s p12s TERM 
          ELEM LECT e1 e3 TERM 
OPTI NOTE LOG 1 
     CSTA 0.5 
CALC TINI 0.0 PAS1 1.E-5 TFIN 5.E-3 
*================================================================= 
PLAY 
CAME   1 EYE   1.00000E+00  5.00000E-01  5.03115E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 

         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.48819E+01 
sler cam1 1 nfra 1 
scen geom navi free 
          face hfro 
     vect scco fiel vite scal user prog 10 pas 10 140 term 
          SUPP LECT air  TERM 
     text vsca 
     colo pape 
freq 0 tfre 2.5e-3 
go 
trac offs fich bmp rend 
go 
trac offs fich bmp rend 
ENDPLAY 
*================================================================= 
SUIT 
Post treatment 
ECHO 
 conv win 
RESU ALIC GARD PSCR 
OPTI PRIN 
SORT VISU NSTO 1 
*================================================================= 
PLAY 
CAME   1 EYE   1.00000E+00  5.00000E-01  5.03115E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.48819E+01 
sler cam1 1 nfra 1 
scen geom navi free 
     iso  fili fiel ecro 1 scal user prog 0.61E5 pas 0.2E5 3.21E5 term 
          SUPP LECT flui TERM 
     text isca 
     vect scco fiel vite scal user prog 10 pas 10 140 term 
          SUPP LECT flui TERM 
     text vsca 
     colo pape 
trac offs fich avi     nocl nfto 101 fps 10 kfre 10 comp -1 rend 
freq 1 
gotr loop 99 offs fich avi cont nocl rend 

go 
trac offs fich avi cont rend 
ENDPLAY 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR  3 'p_e1'    ECRO COMP 1 ELEM LECT e1 TERM 
COUR  4 'p_e3'    ECRO COMP 1 ELEM LECT e3 TERM 

COUR  5 'dx_p10s' DEPL COMP 1 POIN LECT p10s TERM 
COUR  6 'dy_p12s' DEPL COMP 2 POIN LECT p12s TERM 
COUR  7 'dx_p9s'  DEPL COMP 1 POIN LECT p9s TERM 
COUR  8 'dy_p9s'  DEPL COMP 2 POIN LECT p9s TERM 
* 
TRAC 3 4 AXES 1.0 'PRESS [PA]' 
TRAC 5 6 AXES 1.0 'DISPL. [M]' 
TRAC 7 8 AXES 1.0 'DISPL. [M]' 
LIST 3 4 AXES 1.0 'PRESS [PA]' 
LIST 5 6 AXES 1.0 'DISPL. [M]' 
LIST 7 8 AXES 1.0 'DISPL. [M]' 
* 
QUAL ECRO COMP 1 LECT e1 TERM REFE  1.63342E+5 TOLE 5.E-3 
     ECRO COMP 1 LECT e3 TERM REFE  2.22104E+5 TOLE 5.E-3 
*================================================================= 
FIN 

 
Numerical Solutions (Node-Centred Finite Volumes) 
 
VFNCA1 
 
The mesh generation file (K2000) is: 
 
opti echo 1; 
opti titr 'VFNCA1'; 
opti sauv form 'vfnca1.msh'; 
opti trac psc ftra 'vfnca1_mesh.ps'; 
* 
opti dime 2 elem qua4; 
* 
p1 = 0 0; 
p2 = 2 0; 
p3 = 2 1; 
p4 = 0 1; 
p5 = 0.4 0; 
p6 = 0.4 0.4; 
p7 = 0 0.4; 
p8 = 0.4 1; 
* 
tol = 0.001; 
* 
c1 = p1 d 4 p5; 
c2 = p5 d 4 p6; 
c3 = p6 d 4 p7; 
c4 = p7 d 4 p1; 
expl = dall c1 c2 c3 c4 plan; 
* 
p9l = 1 0; 
p9r = 1 0; 
p10 = 1 0.4; 
p11 = 2 0.4; 
* 
c1 = p5 d 6 p9l; 
c2 = p9l d 4 p10; 
c3 = p10 d 6 p6; 
c4 = p6 d 4 p5; 
air1a = dall c1 c2 c3 c4 plan; 
* 
c1 = p9r d 10 p2; 
c2 = p2 d 4 p11; 
c3 = p11 d 10 p10; 
c4 = p10 d 4 p9r; 
air1b = dall c1 c2 c3 c4 plan; 
* 
c1 = p6 d 6 p10 d 10 p11; 
c2 = p11 d 6 p3; 
c3 = p3 d 16 p8; 
c4 = p8 d 6 p6; 
air1c = dall c1 c2 c3 c4 plan; 
* 
elim tol (air1a et air1c); 
elim tol (air1b et air1c); 

air1 = air1a et air1b et air1c; 
* 
c1 = p7 d 4 p6; 
c2 = p6 d 6 p8; 
c3 = p8 d 4 p4; 
c4 = p4 d 6 p7; 
air2 = dall c1 c2 c3 c4 plan; 
elim tol (air1c et air2); 
elim tol (expl et air2); 
elim tol (expl et air1a); 
* 
air = air1 et air2; 
flui = expl et air; 
* 
e1 = expl elem cont p1; 
e3 = air1 elem cont p3; 
* 
p1s = p1 plus p1; 
p2s = p2 plus p1; 
p3s = p3 plus p1; 
p4s = p4 plus p1; 
p9s = p9l plus p1; 
p10s = p10 plus p1; 
p12s = 1 1; 
c1s = p1s d 10 p9s d 10 p2s; 
c2s = p2s d 10 p3s; 
c3s = p3s d 10 p12s d 10 p4s; 
c4s = p4s d 10 p1s; 
c5s = p9s d 4 p10s; 
stru = c1s et c2s et c3s et c4s et c5s; 
* 
c1p = p2s d 10 p9s d 10 p1s; 
c2p = p3s d 10 p2s; 
c3p = p4s d 10 p12s d 10 p3s; 
c4p = p1s d 10 p4s; 
pext = c4p et c3p et c2p et c1p; 
elim tol (stru et pext); 
* 
nfsa = cont flui; 
* 
mesh = flui et stru et pext et e1 et e3 et nfsa; 
* 
sauv form mesh; 
trac qual mesh; 
* 
list (nbel mesh); 
list (nbno mesh); 
* 
fin; 

 
The EUROPLEXUS input file reads: 
 
VFNCA1 
ECHO 
!conv win 
CAST mesh 
DPLA ALE 
DIME 
 PT2L 235 PT3L 64 MC24 200 ED01 64 CL22 60 ZONE 3 
 NALE 26 NBLE 167 NDVC 868 
TERM 
GEOM MC24 flui ED01 stru CL22 pext TERM 
COMP EPAI 0.01 LECT stru TERM 
     COUL turq LECT air  TERM 
          roug LECT expl TERM 
          bleu LECT stru TERM 
          jaun LECT pext TERM 
GRIL LAGR LECT stru TERM 
     EULE LECT nfsa TERM 
     AUTO AUTR 
MATE MCGP NCOM 1 R 8.3143E3 
          COMP 'Air' PM 29.0 CV1 2.07585E4 CV2 0 CV3 0 
          LECT flui TERM 
     VM23 RO 7800. YOUNG 1.6E11 NU 0.333 ELAS 1.05E8 
          TRAC 2 1.05E8 .656256E-3 1.6105E10 1.00066 
          LECT stru TERM 
     IMPE PIMP RO 1.1897 PRES 1.E5 PREF 0 
          LECT pext TERM 
INIT MCOM COMP 'Air' MFRA 1.0 LECT flui TERM 
          PRES 1.E6 LECT expl TERM 
          PRES 1.E5 LECT air  TERM 
          TEMP 586.36 LECT expl TERM 
          TEMP 293.16 LECT air  TERM 
          VEL1 0.0 LECT flui TERM 
          VEL2 0.0 LECT flui TERM 
          VEL3 0.0 LECT flui TERM 
LINK COUP FSA LECT nfsa TERM 
ECRI DEPL VITE ACCE FINT FEXT CONT ECRO MCVA TFRE 1.E-3 
          POIN LECT p1 p5 p6 p7 p3 TERM 
          ELEM LECT e1 e3 TERM 
     FICH ALIC TFRE 5.E-5 
     FICH ALIC TEMP FREQ 1 
          POIN LECT p1 p4 p5 p6 p7 p3 p1s p2s p3s p4s p9s p10s p12s TERM 
          ELEM LECT e1 e3 TERM 
OPTI NOTE LOG 1 
     CSTA 0.5 
     MC ORDR 2 NUFL ROE 
CALC TINI 0.0 PAS1 1.E-5 
     TFIN 5.E-3 
*================================================================= 
PLAY 
CAME   1 EYE   1.00000E+00  5.00000E-01  5.03115E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.48819E+01 
sler cam1 1 nfra 1 
scen geom navi free 
          face hfro 

     vect scco fiel vite scal user prog 10 pas 10 140 term 
          SUPP LECT air  TERM 
     text vsca 
     colo pape 
freq 0 tfre 2.5e-3 
go 
trac offs fich bmp rend 
go 
trac offs fich bmp rend 
ENDPLAY 
*================================================================= 
SUIT 
Post treatment 
ECHO 
 conv win 
RESU ALIC GARD PSCR 
OPTI PRIN 
SORT VISU NSTO 1 
*================================================================= 
PLAY 
CAME   1 EYE   1.00000E+00  5.00000E-01  5.03115E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.48819E+01 
sler cam1 1 nfra 1 
scen geom navi free 
     iso  fili fiel mcpr scal user prog 0.61E5 pas 0.2E5 3.21E5 term 
          SUPP LECT flui TERM 
     text isca 
     vect scco fiel vite scal user prog 10 pas 10 140 term 
          SUPP LECT flui TERM 
     text vsca 
     colo pape 
trac offs fich avi     nocl nfto 101 fps 10 kfre 10 comp -1 rend 
freq 1 
gotr loop 99 offs fich avi cont nocl rend 
go 
trac offs fich avi cont rend 
ENDPLAY 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR  3 'p_e1'    MCPR COMP 1 NOEU LECT p1 TERM 
COUR  4 'p_e3'    MCPR COMP 1 NOEU LECT p3 TERM 
COUR  5 'dx_p10s' DEPL COMP 1 POIN LECT p10s TERM 
COUR  6 'dy_p12s' DEPL COMP 2 POIN LECT p12s TERM 
COUR  7 'dx_p9s'  DEPL COMP 1 POIN LECT p9s TERM 
COUR  8 'dy_p9s'  DEPL COMP 2 POIN LECT p9s TERM 

COUR 101 'p_p4'    MCPR COMP 1 NOEU LECT p4 TERM 
COUR 102 'r_p4'    MCRO COMP 1 NOEU LECT p4 TERM 
COUR 103 't_p4'    MCTE COMP 1 NOEU LECT p4 TERM 
COUR 104 'y1_p4'   MCMF COMP 1 NOEU LECT p4 TERM 
* 
RCOU 13 'p_e1' FICH 'vfncan.pun' RENA 'p_e1_an' 
RCOU 14 'p_e3' FICH 'vfncan.pun' RENA 'p_e3_an' 
RCOU 15 'dx_p10s' FICH 'vfncan.pun' RENA 'dx_p10s_an' 
RCOU 16 'dy_p12s' FICH 'vfncan.pun' RENA 'dy_p12s_an' 
RCOU 17 'dx_p9s' FICH 'vfncan.pun' RENA 'dx_p9s_an' 
RCOU 18 'dy_p9s' FICH 'vfncan.pun' RENA 'dy_p9s_an' 
* 
TRAC 3 4 13 14 AXES 1.0 'PRESS [PA]' 
     COLO noir noir roug roug 
TRAC 5 6 15 16 AXES 1.0 'DISPL. [M]' 

     COLO noir noir roug roug 
TRAC 7 8 17 18 AXES 1.0 'DISPL. [M]' 
     COLO noir noir roug roug 
TRAC 101 AXES 1.0 'PRESS. [PA]' 
TRAC 102 AXES 1.0 'DENS. [KG/M3]' 
TRAC 103 AXES 1.0 'TEMP. [K]' 
TRAC 104 AXES 1.0 'FRAC. [-]' 
LIST 3 4 AXES 1.0 'PRESS [PA]' 
LIST 5 6 AXES 1.0 'DISPL. [M]' 
LIST 7 8 AXES 1.0 'DISPL. [M]' 
* 
QUAL MCPR COMP 1 LECT p1 TERM REFE  1.56019E+5 TOLE 5.E-3 
     MCPR COMP 1 LECT p3 TERM REFE  3.16715E+5 TOLE 5.E-3 
*================================================================= 
FIN 

 
Numerical Solutions (Cell-Centred Finite Volumes) 
 
VFCC00 
 
The mesh generation file (K2000) is: 
 
opti echo 1; 
opti titr 'VFCC00'; 
opti sauv form 'vfcc00.msh'; 
opti trac psc ftra 'vfcc00_mesh.ps'; 
* 
opti dime 2 elem qua4; 
* 
p1 = 0 0; 
p2 = 2 0; 
p3 = 2 1; 
p4 = 0 1; 
p5 = 0.4 0; 
p6 = 0.4 0.4; 
p7 = 0 0.4; 
p8 = 0.4 1; 
* 
tol = 0.001; 
* 
c1 = p1 d 4 p5; 
c2 = p5 d 4 p6; 
c3 = p6 d 4 p7; 
c4 = p7 d 4 p1; 
expl = dall c1 c2 c3 c4 plan; 
* 
c1 = p5 d 16 p2; 
c2 = p2 d 10 p3; 
c3 = p3 d 16 p8; 
c4 = p8 d 6 p6 d 4 p5; 
air1 = dall c1 c2 c3 c4 plan; 
* 
c1 = p7 d 4 p6; 
c2 = p6 d 6 p8; 
c3 = p8 d 4 p4; 
c4 = p4 d 6 p7; 
air2 = dall c1 c2 c3 c4 plan; 

* 
air = air1 et air2; 
flui = expl et air; 
elim tol flui; 
* 
e1 = expl elem cont p1; 
e3 = air1 elem cont p3; 
* 
p1s = p1 plus p1; 
p2s = p2 plus p1; 
p3s = p3 plus p1; 
p4s = p4 plus p1; 
p9s = 1 0; 
p10s = 1 0.4; 
p12s = 1 1; 
c1s = p1s d 10 p9s d 10 p2s; 
c2s = p2s d 10 p3s; 
c3s = p3s d 10 p12s d 10 p4s; 
c4s = p4s d 10 p1s; 
c5s = p9s d 4 p10s; 
stru = c1s et c2s et c3s et c4s et c5s; 
* 
elim tol (flui et stru); 
* 
mesh = flui et stru et e1 et e3; 
* 
sauv form mesh; 
trac qual mesh; 
* 
list (nbel mesh); 
list (nbno mesh); 
* 
fin; 
 

 
The EUROPLEXUS input file reads: 
 
VFCC00 
ECHO 
!conv win 
CAST mesh 
DPLA ALE 
DIME 
 PT2L 231 PT3L 64 Q4VF 200 ED01 64 ZONE 2 
 NALE 26 NBLE 167 
TERM 
GEOM Q4VF flui ED01 stru TERM 
COMP EPAI 0.01 LECT stru TERM 
     COUL turq LECT air  TERM 
          roug LECT expl TERM 
          bleu LECT stru TERM 
GRIL LAGR LECT stru TERM 
     AUTO AUTR 
MATE GAZP RO 5.9485 GAMMA 1.4 CV 716.75 PINI 1.E6 PREF 1.E5 
          LECT expl TERM 
     GAZP RO 1.1897 GAMMA 1.4 CV 716.75 PINI 1.E5 PREF 1.E5 
          LECT air TERM 
     VM23 RO 7800. YOUNG 1.6E11 NU 0.333 ELAS 1.05E8 
          TRAC 2 1.05E8 .656256E-3 1.6105E10 1.00066 
          LECT stru TERM 
ECRI DEPL VITE ACCE FINT FEXT CONT ECRO TFRE 1.E-3 
          POIN LECT p1 p5 p6 p7 p3 TERM 
          ELEM LECT e1 e3 TERM 
     FICH ALIC TFRE 5.E-5 
     FICH ALIC TEMP FREQ 1 
          POIN LECT p1 p5 p6 p7 p3 p1s p2s p3s p4s p9s p10s p12s TERM 
          ELEM LECT e1 e3 TERM 
OPTI NOTE LOG 1 
     CSTA 0.5 
     VFCC FCON 1 ! rusanov   
CALC TINI 0.0 PAS1 1.E-5 TFIN 5.E-3 
*================================================================= 
PLAY 
CAME   1 EYE   1.00000E+00  5.00000E-01  5.03115E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.48819E+01 
sler cam1 1 nfra 1 
scen geom navi free 
          face hfro 
     vect scco fiel vcvi scal user prog 10 pas 10 140 term 
          SUPP LECT air  TERM 
     text vsca 
     colo pape 
freq 0 tfre 2.5e-3 
go 
trac offs fich bmp rend 
go 
trac offs fich bmp rend 
ENDPLAY 
*================================================================= 
SUIT 
Post treatment 
ECHO 
 conv win 
RESU ALIC GARD PSCR 

OPTI PRIN 
SORT VISU NSTO 1 
*================================================================= 
PLAY 
CAME   1 EYE   1.00000E+00  5.00000E-01  5.03115E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.48819E+01 
sler cam1 1 nfra 1 
scen geom navi free 
     iso  fili fiel ecro 1 scal user prog 0.61E5 pas 0.2E5 3.21E5 term 
          SUPP LECT flui TERM 
     text isca 
     vect scco fiel vcvi scal user prog 10 pas 10 140 term 
          SUPP LECT flui TERM 
     text vsca 
     colo pape 
trac offs fich avi     nocl nfto 101 fps 10 kfre 10 comp -1 rend 
freq 1 
gotr loop 99 offs fich avi cont nocl rend 
go 
trac offs fich avi cont rend 
ENDPLAY 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR  1 'vcvi_e1' VCVI NORM ELEM LECT e1 TERM 
COUR  2 'vcvi_e3' VCVI NORM ELEM LECT e3 TERM 
COUR  3 'p_e1'    ECRO COMP 1 ELEM LECT e1 TERM 
COUR  4 'p_e3'    ECRO COMP 1 ELEM LECT e3 TERM 
COUR  5 'dx_p10s' DEPL COMP 1 POIN LECT p10s TERM 
COUR  6 'dy_p12s' DEPL COMP 2 POIN LECT p12s TERM 
COUR  7 'dx_p9s'  DEPL COMP 1 POIN LECT p9s TERM 
COUR  8 'dy_p9s'  DEPL COMP 2 POIN LECT p9s TERM 
* 
TRAC 1 2 AXES 1.0 'VELOC. [M/S]' 
TRAC 3 4 AXES 1.0 'PRESS [PA]' 
TRAC 5 6 AXES 1.0 'DISPL. [M]' 
TRAC 7 8 AXES 1.0 'DISPL. [M]' 
LIST 1 2 AXES 1.0 'VELOC. [M/S]' 
LIST 3 4 AXES 1.0 'PRESS [PA]' 
LIST 5 6 AXES 1.0 'DISPL. [M]' 
LIST 7 8 AXES 1.0 'DISPL. [M]' 
* 
! J'ai l'impression che la qualification de VCVi ne marche pas ... 
!QUAL VCVI COMP 1 LECT e1 TERM REFE  0.00000E+0 TOLE 5.E-3 
!     VCVI COMP 2 LECT e3 TERM REFE  0.00000E+0 TOLE 5.E-3 
QUAL ECRO COMP 1 LECT e1 TERM REFE  1.72471E+5 TOLE 5.E-3 
     ECRO COMP 1 LECT e3 TERM REFE  2.18842E+5 TOLE 5.E-3 
*================================================================= 
FIN

Comparison of Numerical Solutions 
 
EFVFA1 
 
The EUROPLEXUS input file is: 
 
EFVFA1 
ECHO 
* 
RESU ALIC TEMP 'vfcc00.alt' GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR 33 'p_e1_cc'    ECRO COMP 1 ELEM LECT e1 TERM 
COUR 34 'p_e3_cc'    ECRO COMP 1 ELEM LECT e3 TERM 
COUR 35 'dx_p10s_cc' DEPL COMP 1 POIN LECT p10s TERM 
COUR 36 'dy_p12s_cc' DEPL COMP 2 POIN LECT p12s TERM 
COUR 37 'dx_p9s_cc'  DEPL COMP 1 POIN LECT p9s TERM 
COUR 38 'dy_p9s_cc'  DEPL COMP 2 POIN LECT p9s TERM 
* 
RCOU 13 'p_e1'    FICH 'elfi00.pun' RENA 'p_e1_ef' 
RCOU 14 'p_e3'    FICH 'elfi00.pun' RENA 'p_e3_ef' 
RCOU 15 'dx_p10s' FICH 'elfi00.pun' RENA 'dx_p10s_ef' 

RCOU 16 'dy_p12s' FICH 'elfi00.pun' RENA 'dy_p12s_ef' 
RCOU 17 'dx_p9s'  FICH 'elfi00.pun' RENA 'dx_p9s_ef' 
RCOU 18 'dy_p9s'  FICH 'elfi00.pun' RENA 'dy_p9s_ef' 
* 
RCOU 23 'p_e1'    FICH 'vfnca1.pun' RENA 'p_e1_nc' 
RCOU 24 'p_e3'    FICH 'vfnca1.pun' RENA 'p_e3_nc' 
RCOU 25 'dx_p10s' FICH 'vfnca1.pun' RENA 'dx_p10s_nc' 
RCOU 26 'dy_p12s' FICH 'vfnca1.pun' RENA 'dy_p12s_nc' 
RCOU 27 'dx_p9s'  FICH 'vfnca1.pun' RENA 'dx_p9s_nc' 
RCOU 28 'dy_p9s'  FICH 'vfnca1.pun' RENA 'dy_p9s_nc' 
* 
TRAC 13 14 23 24 33 34 AXES 1.0 'PRESS [PA]' 
     COLO noir noir roug roug vert vert 
TRAC 15 16 25 26 35 36 AXES 1.0 'DISPL. [M]' 
     COLO noir noir roug roug vert vert 
TRAC 17 18 27 28 37 38 AXES 1.0 'DISPL. [M]' 
     COLO noir noir roug roug vert vert 
*================================================================ 

 
The initial conditions of the gases in the Finite Element model (FLUT material) are: 
 

 /P Vc cγ = ρ i  p  

Low-pressure 1.4 1.1897 Kg/m3 2.10137E5 J/Kg 1.0E5 Pa 
High-pressure 1.4 5.9485 Kg/m3 4.20274E5 J/Kg 1.0E6 Pa 

 
The FE model is built, as usual, by keeping the fluid and structural nodes distinct 
(although superposed). All fluid nodes along the F-S interface are subjected to the 
LINK FSA condition. 
 
The Node-Centered Finite Volume input mesh is basically the same as for the FE 
model, since the N-C volumes are built up automatically by the code. The only 
notable difference is the presence of an additional layer of CL22 boundary condition 
elements placed along the external envelope of the structure (merged nodes with the 
structure). These elements are used to simulate the presence of the external 
atmosphere to the structure, by means of a constant imposed pressure of 1.0E5 Pa 
(assigned via the IMPE PIMP material model). In the FE (and in the C-C FV) model 
this is not necessary because the fluid material (FLUT or GAZP, respectively) admits a 
PREF parameter (reference pressure) which can be set to 1.0E5 Pa. However, 
unfortunately this does not seem to be possible with the MCGP material used by N-C 
Finite Volumes. 
 
The initial conditions of the gases in the Node-Centered Finite Volume model are: 
 

w  R Vc  
29 Kg/Kmol 8314.3 J/Kmol·K 20785.75 J/ Kmol·K 

 
and: 
 

 T p
Low-pressure 293.16 K 1.0E5 Pa
High-pressure 586.36 K 1.0E6 Pa

 
As concerns fluid-structure interaction, the same LINK FSA condition as used in the 
FE model is repeated here. 

The Cell-Centered Finite Volume input mesh is similar to the FE model, but the 
structure and fluid nodes are merged together (ELIM command in Cast3m). No FSA 
condition is needed in this case. 
 
The initial conditions of the gases in the Cell-Centered Finite Volume model (GAZP 
material) are similar to those for FE (FLUT material): 
 

 /P Vc cγ = ρ
Vc  p  

Low-pressure 1.4 1.1897 Kg/m3 716.75 J/ Kg·K 1.0E5 Pa 
High-pressure 1.4 5.9485 Kg/m3 716.75 J/ Kg·K 1.0E6 Pa 

 
Note that the initial pressure is given directly rather than specifying the specific 
internal energy. Furthermore, the (optional) specific heat at constant volume Vc  is 
specified. This has no effect on the solution, but is used to compute and print the 
temperatures. 
 
As already mentioned, both in the FE and in the C-C FV models a reference pressure 
value of PREF 1.E5 is specified for both gases. 
Some results: fluid pressures: 
 
 

 
 

 
 
 
 
 



 7

Structural displacements: 
 

 
 
 
 

Structural displacements in another point: 

Final solution with Finite Elements and with Node-Centred Finite Volumes: 
 

 
 

Final solution with Cell-Centred Finite Volumes: 

 


